Pensieve header: The full sl_2 invariant using the Drinfel'd double. Continues 2018-05/ybax.nb, Talks/StonyBrook-1805/ybax.nb, Projects/SL2Portfolio/Logoi.nb. (Last version before "categorification" of 𝔼.

Profiling

In[∘]:=
```
Once[
  SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Invariant"];
  << KnotTheory`;
  << "../Profile/Profile.m";
 ];
BeginProfile[];
Once@PopupWindow[Button["Show Profile Monitor"],
  Dynamic[PrintProfile[], UpdateInterval → 3, TrackedSymbols → {}]]
```

```
Loading KnotTheory` version of January 20, 2015, 10:42:19.1122.
Read more at http://katlas.org/wiki/KnotTheory.
This is Profile.m of http://www.drorbn.net/AcademicPensieve/Projects/Profile/.
This version: June 2018. Original version: July 1994.
```

Out[∘]= Show Profile Monitor

External Utilities

In[∘]:=
```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Program

Program

Internal Utilities

Program

Canonical Form:

Program

In[∘]:=
```
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := PP_CF@ExpandDenominator@ExpandNumerator@PP_Together@Together[PP_Exp[
     Expand[ℰ] //. ⅇ^x_ ⅇ^y_ :> ⅇ^(x+y) /. ⅇ^x_ :> ⅇ^CF[x]]];
```

Program

The Kronecker δ:

Program

In[∘]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Program

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

Program

```
In[◦]:=
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]$k_ := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

Zip and Bind

Program

Variables and their duals:

Program

```
In[◦]:=
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_i_)* := (u*)i;
```

Program

Finite Zips:

Program

```
In[◦]:=
collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := PPCollect@Collect[ℰ, ζ];
Zip{}[P_] := P; Zip{ζ_,ζs___}[P_] := PPZip[
  (collect[P // Zip{ζs}, ζ] /. f_. ζ^d_. :> ∂{ζ*,d} f) /. ζ* → 0]
```

Program

QZip implements the “Q-level zips” on $\mathbb{E}(L, Q, P) = P e^{L+Q}$. Such zips regard the L variables as scalars.

Program

```
In[◦]:=
QZipζs_List@𝔼[L_, Q_, P_] := PPQZip@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = Inverse@Table[Kδz,ζ* - ∂z,ζ Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → qt.(zs + ys)];
    Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    CF /@ 𝔼[L, Q2, Det[qt] e^-Q2 Zipζs[e^Q1 (P /. zrule)]] ];
```

Program

Upper to lower and lower to Upper:

Program

```
In[◦]:=
U2l = {B_i_^p_. → e^(-p ℏ γ b_i), B^p_. → e^(-p ℏ γ b), T_i_^p_. → e^(p ℏ t_i), T^p_. → e^(p ℏ t), 𝒜_i_^p_. → e^(p γ α_i), 𝒜^p_. → e^(p γ α)};
l2U = {e^(c_. b_i_ + d_.) :> B_i^(-c/(ℏ γ)) e^d, e^(c_. b + d_.) :> B^(-c/(ℏ γ)) e^d,
    e^(c_. t_i_ + d_.) :> T_i^(c/ℏ) e^d, e^(c_. t + d_.) :> T^(c/ℏ) e^d,
    e^(c_. α_i_ + d_.) :> 𝒜_i^(c/γ) e^d, e^(c_. α + d_.) :> 𝒜^(c/γ) e^d,
    e^ℰ_ :> e^(Expand@ℰ)};
```

Program

LZip implements the “L-level zips” on $\mathbb{E}(L, Q, P) = P e^{L+Q}$. Such zips regard all of $P e^Q$ as a single”P”. Here the z’s are b and α and the ζ’s are β and a.

Program

```
In[•]:= LZip_{𝔷s_List}@𝔼[L_, Q_, P_] := PP_LZip@Module[{𝔷, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
    zs = Table[𝔷*, {𝔷, 𝔷s}];
    c = L /. Alternatives @@ (𝔷s ⋃ zs) → 0;
    ys = Table[∂_𝔷 (L /. Alternatives @@ zs → 0), {𝔷, 𝔷s}];
    ηs = Table[∂_z (L /. Alternatives @@ 𝔷s → 0), {z, zs}];
    lt = Inverse@Table[Kδ_{z,𝔷*} - ∂_{z,𝔷} L, {𝔷, 𝔷s}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
    CF /@ 𝔼[L2, Q2, Det[lt] ⅇ^{-L2-Q2} Zip_{𝔷s}[ⅇ^{L1+Q1} (P /. U2l /. zrule)]] //. l2U];
```

Program

```
In[•]:= B_{}[L_, R_] := L R;
B_{is__}[L_𝔼, R_𝔼] := PP_B@Module[{n},
    Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Join@@Table[{β_{n@i}, τ_{n@i}, a_{n@i}}, {i, {is}}]} // QZip_{Join@@Table[{ξ_{n@i}, y_{n@i}}, {i, {is}}]} ];
B_{is___}[L_, R_] := B_{{is}}[L, R];
```

Program

“Define” code

Program

Define[lhs = rhs, ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

```
In[•]:= SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
Define[op_{_is__} = ε_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
   ReleaseHold[Hold[
      SD[op_{nisp,$k_Integer}, PP_{Boot@$k}@Block[{i, j, k}, op_{isp,$k} = ε; op_{nis,$k}]];
      SD[op_{isp}, op_{{is},$k}]; SD[op_{sis__}, op_{{sis}}];
     ] /. {SD → SetDelayed,
      isp → {is} /. {i → i_, j → j_, k → k_},
      nis → {is} /. {i → ii, j → jj, k → kk},
      nisp → {is} /. {i → ii_, j → jj_, k → kk_}
     }] ]]
```

Program

Booting Up

Program

```
In[•]:= $k = 2; (*ℏ=γ=1;*)
```

Program

In[∘]:=
```
Define[am_{i,j→k} = 𝔼[(α_i + α_j) a_k, (e^{-γ α_j} ξ_i + ξ_j) x_k, 1]_{$k},
  bm_{i,j→k} = 𝔼[(β_i + β_j) b_k, (η_i + η_j) y_k, e^{(e^{-ϵ β_i} - 1) η_j y_k}]_{$k}]
```

Program

In[∘]:=
```
Define[R_{i,j} = 𝔼[ℏ a_j b_i, ℏ x_j y_i, e^(Sum_{k=2}^{$k+1} ((1 - e^{γ ϵ ℏ})^k (ℏ y_i x_j)^k) / (k (1 - e^{k γ ϵ ℏ})))]_{$k},
  R̄_{i,j} = 𝔼[-ℏ a_j b_i, -ℏ x_j y_i / B_i, 1 + If[$k == 0, 0, (R̄_{{i,j},$k-1})_{$k}⟦3⟧ -
      (((R̄_{{i,j},0})_{$k} R_{1,2} (R̄_{{3,4},$k-1})_{$k}) ~ B_{i,j,1,2} ~ (bm_{i,1→i} am_{j,2→j}) ~ B_{i,j,3,4} ~ (bm_{i,3→i} am_{j,4→j}))⟦3⟧]],
  P_{i,j} = 𝔼[β_i α_j / ℏ, η_i ξ_j / ℏ, 1 + If[$k == 0, 0, (P_{{i,j},$k-1})_{$k}⟦3⟧ -
      (R_{1,2} ~ B_{1,2} ~ ((P_{{1,j},0})_{$k} (P_{{i,2},$k-1})_{$k}))⟦3⟧]]]
```

Program

In[∘]:=
```
Define[aS_j = R̄_{i,j} ~ B_i ~ P_{i,j},
  aS̄_i = 𝔼[-a_i α_i, -x_i 𝒜_i ξ_i, 1 + If[$k == 0, 0, (aS̄_{{i},$k-1})_{$k}⟦3⟧ -
      ((aS̄_{{i},0})_{$k} ~ B_i ~ aS_i ~ B_i ~ (aS̄_{{i},$k-1})_{$k})⟦3⟧]]]
```

Program

In[∘]:=
```
Define[bS_i = R_{i,1} ~ B_1 ~ aS_1 ~ B_1 ~ P_{i,1},
  bS̄_i = R_{i,1} ~ B_1 ~ aS̄_1 ~ B_1 ~ P_{i,1},
  aΔ_{i→j,k} = (R_{1,j} R_{2,k}) ~ B_{1,2} ~ bm_{1,2→3} ~ B_3 ~ P_{3,i},
  bΔ_{i→j,k} = (R_{j,1} R_{k,2}) ~ B_{1,2} ~ am_{1,2→3} ~ B_3 ~ P_{i,3}]
```

Program

In[∘]:=
```
Define[dm_{i,j→k} =
    (𝔼[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1] (aΔ_{i→1,2} ~ B_2 ~ aΔ_{2→2,3} ~ B_3 ~ aS̄_3) (bΔ_{j→-1,-2} ~ B_{-2} ~ bΔ_{-2→-2,-3})) ~
    B_{-3,-2,-1,1,2,3,i,j} ~ (P_{-1,3} P_{-3,1} am_{2,j→k} bm_{i,-2→k}),
  dS_i = 𝔼[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] ~ B_{1,2} ~ (bS̄_1 aS_2) ~ B_{1,2} ~ dm_{2,1→i},
  dΔ_{i→j,k} = (bΔ_{i→3,1} aΔ_{i→2,4}) ~ B_{1,2,3,4} ~ (dm_{3,4→k} dm_{1,2→j})]
```

Program

In[∘]:=
```
Define[C_i = 𝔼[0, 0, B_i^{1/2} e^{-ℏ ϵ a_i/2}]_{$k},
  C̄_i = 𝔼[0, 0, B_i^{-1/2} e^{ℏ ϵ a_i/2}]_{$k},
  Kink_i = (R_{1,3} C̄_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i},
  K̄ink_i = (R̄_{1,3} C_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i}]
```

Program

Note. $t == \epsilon a - \gamma b$ and $b == -t/\gamma + \epsilon a/\gamma$.

Program

In[∘]:=
```
Define[b2t_i = 𝔼[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, e^{ϵ β_i a_i/γ}]_{$k},
  t2b_i = 𝔼[α_i a_i - τ_i γ b_i, ξ_i x_i + η_i y_i, e^{ϵ τ_i a_i}]_{$k}]
```

Testing

In[∘]:=
```
Block[{$k = 1}, {
    am → am_{i,j→k}, bm → bm_{i,j→k}, dm → dm_{i,j→k}, R → R_{i,j}, R̄ → R̄_{i,j}, P → P_{i,j},
    aS → aS_i, aS̄ → aS̄_i, bS → bS_i, bS̄ → bS̄_i, dS → dS_i, aΔ → aΔ_{i→j,k}, bΔ → bΔ_{i→j,k},
    dΔ → dΔ_{i→j,k}, C → C_i, C̄ → C̄_i, Kink → Kink_i, Kink̄ → Kink̄_i, b2t → b2t_i, t2b → t2b_i
  }] //
 Column
```

Out[∘]=

$$am \to \mathbb{E}\left[a_k\,(\alpha_i+\alpha_j),\ x_k\left(e^{-\gamma\,\alpha_j}\,\xi_i+\xi_j\right),\ 1\right]$$

$$bm \to \mathbb{E}\left[b_k\,(\beta_i+\beta_j),\ y_k\,(\eta_i+\eta_j),\ 1-y_k\,\beta_i\,\eta_j\,\epsilon+O[\epsilon]^2\right]$$

$$dm \to \mathbb{E}\Big[a_k\,\alpha_i+a_k\,\alpha_j+b_k\,\beta_i+b_k\,\beta_j,\ \frac{1}{\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\left(\hbar\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_i+\hbar\,y_k\,\mathcal{A}_j\,\eta_j+\hbar\,x_k\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i-B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i+\hbar\,x_k\,\mathcal{A}_i\,\mathcal{A}_j\,\xi_j\right),$$
$$1+\frac{1}{4\,\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\Big(-4\,\hbar\,y_k\,\mathcal{A}_j\,\beta_i\,\eta_j-4\,\hbar\,x_k\,\mathcal{A}_i\,\beta_j\,\xi_i+4\,\gamma\,\hbar^2\,x_k\,y_k\,\eta_j\,\xi_i+4\,\hbar\,a_k\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i+2\,\gamma\,\hbar\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i-6\,\gamma\,\hbar\,B_k\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i+2\,\gamma\,\hbar\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2-$$
$$6\,\gamma\,\hbar\,B_k\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2+\gamma\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2-4\,\gamma\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2+3\,\gamma\,B_k^2\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2\Big)\,\epsilon+O[\epsilon]^2\Big]$$

$$R \to \mathbb{E}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1-\frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{R} \to \mathbb{E}\left[-\hbar\,a_j\,b_i,\ -\frac{\hbar\,x_j\,y_i}{B_i},\ 1-\frac{\left(4\,\hbar^2\,a_j\,B_i\,x_j\,y_i+3\,\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon}{4\,B_i^2}+O[\epsilon]^2\right]$$

$$P \to \mathbb{E}\left[\frac{\alpha_j\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_j}{\hbar},\ 1+\frac{\gamma\,\eta_i^2\,\xi_j^2\,\epsilon}{4\,\hbar}+O[\epsilon]^2\right]$$

$$aS \to \mathbb{E}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(-2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{aS} \to \mathbb{E}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(2\,\gamma\,\hbar\,x_i\,\mathcal{A}_i\,\xi_i-2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$bS \to \mathbb{E}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\frac{\left(-2\,B_i\,y_i\,\beta_i\,\eta_i-\gamma\,\hbar\,y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2}+O[\epsilon]^2\right]$$

$$\overline{bS} \to \mathbb{E}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\frac{\left(2\,\gamma\,\hbar\,B_i\,y_i\,\eta_i-2\,B_i\,y_i\,\beta_i\,\eta_i-\gamma\,\hbar\,y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2}+O[\epsilon]^2\right]$$

$$dS \to \mathbb{E}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\ \frac{-\hbar\,y_i\,\mathcal{A}_i\,\eta_i-\hbar\,B_i\,x_i\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\eta_i\,\xi_i-B_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{\hbar\,B_i},$$
$$1+\frac{1}{4\,\hbar\,B_i^2}\Big(4\,\gamma\,\hbar^2\,B_i\,y_i\,\mathcal{A}_i\,\eta_i-4\,\hbar\,B_i\,y_i\,\mathcal{A}_i\,\beta_i\,\eta_i-2\,\gamma\,\hbar^2\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2-4\,\hbar^2\,a_i\,B_i^2\,x_i\,\mathcal{A}_i\,\xi_i-4\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i\,\beta_i\,\xi_i-$$
$$4\,\gamma\,\hbar\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\hbar\,a_i\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\gamma\,\hbar\,B_i^2\,\mathcal{A}_i\,\eta_i\,\xi_i-4\,\gamma\,\hbar^2\,B_i\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i+$$
$$4\,B_i\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i-4\,B_i^2\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i+6\,\gamma\,\hbar\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\gamma\,\hbar\,B_i\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\gamma\,\hbar^2\,B_i^2\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+$$
$$6\,\gamma\,\hbar\,B_i\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-2\,\gamma\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-3\,\gamma\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+4\,\gamma\,B_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2-\gamma\,B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2\Big)\,\epsilon+O[\epsilon]^2\Big]$$

$$a\Delta \to \mathbb{E}\left[a_j\,\alpha_i+a_k\,\alpha_i,\ x_j\,\xi_i+x_k\,\xi_i,\ 1+\frac{1}{2}\left(-2\,\hbar\,a_j\,x_k\,\xi_i+\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$b\Delta \to \mathbb{E}\left[b_j\,\beta_i+b_k\,\beta_i,\ B_k\,y_j\,\eta_i+y_k\,\eta_i,\ 1+\frac{1}{2}\,\gamma\,\hbar\,B_k\,y_j\,y_k\,\eta_i^2\,\epsilon+O[\epsilon]^2\right]$$

$$d\Delta \to \mathbb{E}\Big[a_j\,\alpha_i+a_k\,\alpha_i+b_j\,\beta_i+b_k\,\beta_i,\ y_j\,\eta_i+B_j\,y_k\,\eta_i+x_j\,\xi_i+x_k\,\xi_i,$$
$$1+\frac{1}{2}\left(\gamma\,\hbar\,B_j\,y_j\,y_k\,\eta_i^2-2\,\hbar\,a_j\,x_k\,\xi_i+\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

$$C \to \mathbb{E}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(\hbar\,a_i\,\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{C} \to \mathbb{E}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{\hbar\,a_i\,\epsilon}{2\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$Kink \to \mathbb{E}\left[\hbar\,a_i\,b_i,\ \hbar\,x_i\,y_i,\ \frac{1}{\sqrt{B_i}}+\frac{\left(2\,\hbar\,a_i-\gamma\,\hbar^3\,x_i^2\,y_i^2\right)\epsilon}{4\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$\overline{Kink} \to \mathbb{E}\left[-\hbar\,a_i\,b_i,\ -\frac{\hbar\,x_i\,y_i}{B_i},\ \sqrt{B_i}+\frac{\left(-2\,\hbar\,a_i\,B_i^2-4\,\hbar^2\,a_i\,B_i\,x_i\,y_i-3\,\gamma\,\hbar^3\,x_i^2\,y_i^2\right)\epsilon}{4\,B_i^{3/2}}+O[\epsilon]^2\right]$$

$$b2t \to \mathbb{E}\left[a_i\,\alpha_i-\frac{t_i\,\beta_i}{\gamma},\ y_i\,\eta_i+x_i\,\xi_i,\ 1+\frac{a_i\,\beta_i\,\epsilon}{\gamma}+O[\epsilon]^2\right]$$

$$t2b \to \mathbb{E}\left[a_i\,\alpha_i-\gamma\,b_i\,\tau_i,\ y_i\,\eta_i+x_i\,\xi_i,\ 1+a_i\,\tau_i\,\epsilon+O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

```
In[∘]:= Timing@{{"[a,x]" → ((𝔼[0, 0, a₂ x₁] ~B₁,₂~ am₁,₂→₁)⟦3⟧ - (𝔼[0, 0, a₁ x₂] ~B₁,₂~ am₁,₂→₁)⟦3⟧),
      "[b,y]" → ((𝔼[0, 0, y₂ b₁] ~B₁,₂~ bm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, y₁ b₂] ~B₁,₂~ bm₁,₂→₁)⟦3⟧)} /.
    z_1 → z,
   {"Δ[y]" → Last[𝔼[0, 0, y₁] ~B₁~ bΔ₁→₁,₂],
    "Δ[b]" → Last[𝔼[0, 0, b₁] ~B₁~ bΔ₁→₁,₂],
    "Δ[a]" → Last[𝔼[0, 0, a₁] ~B₁~ aΔ₁→₁,₂],
    "Δ[x]" → Last[𝔼[0, 0, x₁] ~B₁~ aΔ₁→₁,₂]},
   {
     "S(a)" → ((𝔼[0, 0, a₁] ~B₁~ aS₁)⟦3⟧),
     "S(x)" → ((𝔼[0, 0, x₁] ~B₁~ aS₁)⟦3⟧),
     "S(b)" → ((𝔼[0, 0, b₁] ~B₁~ bS₁)⟦3⟧),
     "S(y)" → ((𝔼[0, 0, y₁] ~B₁~ bS₁)⟦3⟧)
   } /. z_1 → z}
```

Out[∘]= $\{1.10938, \{\{[a,x] \to -x\gamma,\ [b,y] \to -y\epsilon + O[\epsilon]^3\},\ \{\Delta[y] \to (B_2 y_1 + y_2) + O[\epsilon]^3,\ \Delta[b] \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta[a] \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta[x] \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3\},\ \{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately.

Associativity of am and bm:

```
In[∘]:= Timing@Block[{$k = 3},
   HL /@ {(am₁,₂→₁ ~B₁~ am₁,₃→₁) ≡ (am₂,₃→₂ ~B₂~ am₁,₂→₁), (bm₁,₂→₁ ~B₁~ bm₁,₃→₁) ≡ (bm₂,₃→₂ ~B₂~ bm₁,₂→₁)}
  ]
```

Out[∘]= {0.125, {True, True}}

R and *P* are inverses:

```
In[∘]:= Timing@Block[{$k = 3}, {Rᵢ,ⱼ, Pᵢ,ₖ, HL[Rᵢ,ⱼ ~Bᵢ~ Pᵢ,ₖ ≡ 𝔼[aⱼ αₖ, xⱼ ξₖ, 1]]}]
```

Out[∘]= $\{0.109375, \{\mathbb{E}[\hbar a_j b_i, \hbar x_j y_i, 1 - \frac{1}{4}(\gamma \hbar^3 x_j^2 y_i^2)\epsilon + \left(\frac{1}{9}\gamma^2\hbar^5 x_j^3 y_i^3 + \frac{1}{32}\gamma^2\hbar^6 x_j^4 y_i^4\right)\epsilon^2 + \frac{1}{1152}(24\gamma^3\hbar^5 x_j^2 y_i^2 - 72\gamma^3\hbar^7 x_j^4 y_i^4 - 32\gamma^3\hbar^8 x_j^5 y_i^5 - 3\gamma^3\hbar^9 x_j^6 y_i^6)\epsilon^3 + O[\epsilon]^4],$
$\mathbb{E}[\frac{\alpha_k \beta_i}{\hbar}, \frac{\eta_i \xi_k}{\hbar}, 1 + \frac{\gamma \eta_i^2 \xi_k^2 \epsilon}{4\hbar} + \frac{(36\gamma^2\hbar^2\eta_i^2\xi_k^2 + 40\gamma^2\hbar\eta_i^3\xi_k^3 + 9\gamma^2\eta_i^4\xi_k^4)\epsilon^2}{288\hbar^2} - \frac{1}{1152\hbar^3}(-48\gamma^3\hbar^4\eta_i^2\xi_k^2 - 192\gamma^3\hbar^3\eta_i^3\xi_k^3 - 156\gamma^3\hbar^2\eta_i^4\xi_k^4 - 40\gamma^3\hbar\eta_i^5\xi_k^5 - 3\gamma^3\eta_i^6\xi_k^6)\epsilon^3 + O[\epsilon]^4], \text{True}\}\}$

as and $\overline{\text{aS}}$ are inverses, bs and $\overline{\text{bS}}$ are inverses:

```
In[∘]:= Timing[HL /@ {a̅S̅₁ ~B₁~ aS₁ ≡ 𝔼[a₁ α₁, x₁ ξ₁, 1], b̅S̅₁ ~B₁~ bS₁ ≡ 𝔼[b₁ β₁, y₁ η₁, 1]}]
```

Out[∘]= {0.453125, {True, True}}

(co)-associativity on both sides

```
In[∘]:= Timing[HL /@
  {(aΔ1→1,2 ~ B2 ~ aΔ2→2,3) ≡ (aΔ1→1,3 ~ B1 ~ aΔ1→1,2), (bΔ1→1,2 ~ B2 ~ bΔ2→2,3) ≡ (bΔ1→1,3 ~ B1 ~ bΔ1→1,2),
   (am1,2→1 ~ B1 ~ am1,3→1) ≡ (am2,3→2 ~ B2 ~ am1,2→1), (bm1,2→1 ~ B1 ~ bm1,3→1) ≡ (bm2,3→2 ~ B2 ~ bm1,2→1)}]
Out[∘]= {0.40625, {True, True, True, True}}
```

Δ is an algebra morphism

```
In[∘]:= Timing[HL /@ {am1,2→1 ~ B1 ~ aΔ1→1,2 ≡ (aΔ1→1,3 aΔ2→2,4) ~ B1,2,3,4 ~ (am3,4→2 am1,2→1),
    bm1,2→1 ~ B1 ~ bΔ1→1,2 ≡ (bΔ1→1,3 bΔ2→2,4) ~ B1,2,3,4 ~ (bm3,4→2 bm1,2→1)}]
Out[∘]= {0.546875, {True, True}}
```

An explicit formula for aS_i

```
In[∘]:= Timing@Block[{$k = 4}, HL[aSi ≡ 𝔼[-αi aj, -ξi xi,
        Sum[Expand[(ℯ^(ξi xi) (-ℏ γ ϵ)^k)/(2^k k!) Nest[Expand[xi^2 ∂{xi,2} #] &, ℯ^(-ξi ℯ^(ℏ ϵ ai) xi), k]], {k, 0, $k}]]$k ~
     Bi,j ~ ami,j→i]]
Out[∘]= {3.26563, True}
```

S is convolution inverse of id

```
In[∘]:= Timing[HL[# ≡ 𝔼[0, 0, 1]] & /@ {
    (aΔ1→1,2 ~ B1 ~ aS1) ~ B1,2 ~ am1,2→1, (aΔ1→1,2 ~ B2 ~ aS2) ~ B1,2 ~ am1,2→1,
    (bΔ1→1,2 ~ B1 ~ bS1) ~ B1,2 ~ bm1,2→1, (bΔ1→1,2 ~ B2 ~ bS2) ~ B1,2 ~ bm1,2→1}]
Out[∘]= {0.6875, {True, True, True, True}}
```

But not with the opposite product:

```
In[∘]:= Timing[Short[# ≡ 𝔼[0, 0, 1]] & /@ {
    (aΔ1→1,2 ~ B1 ~ aS1) ~ B1,2 ~ am2,1→1, (aΔ1→1,2 ~ B2 ~ aS2) ~ B1,2 ~ am2,1→1,
    (bΔ1→1,2 ~ B1 ~ bS1) ~ B1,2 ~ bm2,1→1, (bΔ1→1,2 ~ B2 ~ bS2) ~ B1,2 ~ bm2,1→1}]
```

Out[∘]= $\Big\{0.609375, \Big\{\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \mathcal{A}_1 \xi_1 + \gamma^2\epsilon^2\hbar^2 x_1 \mathcal{A}_1 \xi_1 - 2\gamma\epsilon^2\hbar^2 a_1 x_1 \mathcal{A}_1 \xi_1 + 2\gamma^2\epsilon^2\hbar^2 x_1^2 \mathcal{A}_1^2 \xi_1^2\right) == 0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \xi_1 - \gamma^2\epsilon^2\hbar^2 x_1 \xi_1 + 2\gamma^2\epsilon^2\hbar^2 x_1^2 \xi_1^2\right) == 0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar y_1 \eta_1 - \gamma^2\epsilon^2\hbar^2 y_1 \eta_1 + 2\gamma^2\epsilon^2\hbar^2 y_1^2 \eta_1^2\right) == 0,$

$\frac{-2\gamma\epsilon\hbar B_1 y_1 \eta_1 + \ll 3 \gg + 2\gamma^2\epsilon^2\hbar^2 y_1^2 \eta_1^2}{2 B_1^2} == 0\Big\}\Big\}$

S is an algebra anti-(co)morphism

```
In[∘]:= Timing[HL /@ {am1,2→1 ~ B1 ~ aS1 ≡ (aS1 aS2) ~ B1,2 ~ am2,1→1, bm1,2→1 ~ B1 ~ bS1 ≡ (bS1 bS2) ~ B1,2 ~ bm2,1→1,
    aS1 ~ B1 ~ aΔ1→1,2 ≡ aΔ1→2,1 ~ B1,2 ~ (aS1 aS2), bS1 ~ B1 ~ bΔ1→1,2 ≡ bΔ1→2,1 ~ B1,2 ~ (bS1 bS2)}]
Out[∘]= {0.8125, {True, True, True, True}}
```

Pairing axioms

```
In[]:= Timing[HL /@ {(bm1,2→1 𝔼[α3 a3, ξ3 x3, 1]) ~B1,3~P1,3 ≡
         (𝔼[β1 b1, η1 y1, 1] 𝔼[β2 b2, η2 y2, 1] aΔ3→4,5) ~B1,4~P1,4~B2,5~P2,5,
       (bΔ1→1,2 𝔼[α3 a3, ξ3 x3, 1] 𝔼[α4 a4, ξ4 x4, 1]) ~B1,3~P1,3~B2,4~P2,4 ≡
         (𝔼[β1 b1, η1 y1, 1] am3,4→3) ~B1,3~P1,3 }]
Out[]= {0.40625, {True, True}}
```

```
In[]:= Timing[HL /@ {(bS1 𝔼[α2 a2, ξ2 x2, 1]) ~B1,2~P1,2 ≡ (𝔼[β1 b1, η1 y1, 1] aS2) ~B1,2~P1,2,
       (bS̄1 𝔼[α2 a2, ξ2 x2, 1]) ~B1,2~P1,2 ≡ (𝔼[β1 b1, η1 y1, 1] aS̄2) ~B1,2~P1,2}]
Out[]= {0.296875, {True, True}}
```

Tests for the double.

Check the double formulas on the generators agree with SL2Portfolio.pdf:

```
In[]:= Timing@{{
          "[a,y]" → ((𝔼[0, 0, y2 a1] ~B1,2~dm1,2→1)〚3〛 - (𝔼[0, 0, y1 a2] ~B1,2~dm1,2→1)〚3〛),
          "[b,x]"  → ((𝔼[0, 0, x2 b1] ~B1,2~dm1,2→1)〚3〛 - (𝔼[0, 0, x1 b2] ~B1,2~dm1,2→1)〚3〛),
          "xy-qyx" → ((𝔼[0, 0, x1 y2] ~B1,2~dm1,2→1)〚3〛 - (1 + ϵ) (𝔼[0, 0, y1 x2] ~B1,2~dm1,2→1)〚3〛)
        } /. {z_1 → z} // Expand // Factor,
      {
        "Δ(a)" → ((𝔼[0, 0, a1] ~B1~dΔ1→1,2)〚3〛),
        "Δ(x)" → ((𝔼[0, 0, x1] ~B1~dΔ1→1,2)〚3〛),
        "Δ(b)" → ((𝔼[0, 0, b1] ~B1~dΔ1→1,2)〚3〛),
        "Δ(y)" → ((𝔼[0, 0, y1] ~B1~dΔ1→1,2)〚3〛)
      } // Simplify,
      {
         "S(a)" → ((𝔼[0, 0, a1] ~B1~dS1)〚3〛),
         "S(x)" → ((𝔼[0, 0, x1] ~B1~dS1)〚3〛),
         "S(b)" → ((𝔼[0, 0, b1] ~B1~dS1)〚3〛),
         "S(y)" → ((𝔼[0, 0, y1] ~B1~dS1)〚3〛)
      } /. {z_1 → z} // Simplify
    }
```

Out[]=
$$\left\{7.32813, \left\{\left\{[a,y] \to -y\,\gamma + O[\epsilon]^3,\ [b,x] \to x\,\epsilon + O[\epsilon]^3,\right.\right.\right.$$
$$\left. xy{-}qyx \to \left(-x\,y + \frac{1 - B + x\,y\,\hbar}{\hbar}\right) + (a\,B - x\,y + x\,y\,\gamma\,\hbar)\,\epsilon + \frac{1}{2}\left(-a^2\,B\,\hbar + x\,y\,\gamma^2\,\hbar^2\right)\epsilon^2 + O[\epsilon]^3\right\},$$
$$\left\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - \hbar\,a_1\,x_2\,\epsilon + \frac{1}{2}\hbar^2\,a_1^2\,x_2\,\epsilon^2 + O[\epsilon]^3,\right.$$
$$\left.\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1\,y_2) + O[\epsilon]^3\right\},$$
$$\left\{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a\,x\,\hbar\,\epsilon - \frac{1}{2}\left(a^2\,x\,\hbar^2\right)\epsilon^2 + O[\epsilon]^3,\right.$$
$$\left.\left.\left. S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y\,\gamma\,\hbar\,\epsilon}{B} - \frac{\left(y\,\gamma^2\,\hbar^2\right)\epsilon^2}{2\,B} + O[\epsilon]^3\right\}\right\}\right\}$$

(co)-associativity

In[∘]:= Timing[HL /@
 {($d\Delta_{1\to 1,2} \sim B_2 \sim d\Delta_{2\to 2,3}$) ≡ ($d\Delta_{1\to 1,3} \sim B_1 \sim d\Delta_{1\to 1,2}$), ($dm_{1,2\to 1} \sim B_1 \sim dm_{1,3\to 1}$) ≡ ($dm_{2,3\to 2} \sim B_2 \sim dm_{1,2->1}$)}]

Out[∘]= {5.89063, {True, True}}

Δ is an algebra morphism

In[∘]:= Timing@HL[$dm_{1,2\to 1} \sim B_1 \sim d\Delta_{1\to 1,2}$ ≡ ($d\Delta_{1\to 1,3}\ d\Delta_{2\to 2,4}$) ~ $B_{1,2,3,4}$ ~ ($dm_{3,4\to 2}\ dm_{1,2\to 1}$)]

Out[∘]= {6.90625, True}

S_2 inverts R, but not S_1:

In[∘]:= Timing@{$R_{1,2} \sim B_1 \sim dS_1 \equiv \overline{R}_{1,2}$, HL[$R_{1,2} \sim B_2 \sim dS_2 \equiv \overline{R}_{1,2}$]}

Out[∘]= $\left\{0.65625, \left\{\frac{1}{4 B_1^3}\left(4 \gamma \epsilon \hbar^2 B_1^2 x_2 y_1 - 2 \gamma^2 \epsilon^2 \hbar^3 B_1^2 x_2 y_1 + 4 \gamma \epsilon^2 \hbar^3 a_2 B_1^2 x_2 y_1 + 8 \gamma^2 \epsilon^2 \hbar^4 B_1 x_2^2 y_1^2 - 4 \gamma \epsilon^2 \hbar^4 a_2 B_1 x_2^2 y_1^2 - 3 \gamma^2 \epsilon^2 \hbar^5 x_2^3 y_1^3\right) == 0, \text{True}\right\}\right\}$

S is convolution inverse of id

In[∘]:= Timing[
 HL[# ≡ 𝔼[0, 0, 1]] & /@ {($d\Delta_{1\to 1,2} \sim B_1 \sim dS_1$) ~ $B_{1,2}$ ~ $dm_{1,2\to 1}$, ($d\Delta_{1\to 1,2} \sim B_2 \sim dS_2$) ~ $B_{1,2}$ ~ $dm_{1,2\to 1}$}]

Out[∘]= {8.71875, {True, True}}

S is a (co)-algebra anti-morphism

In[∘]:= Timing[HL /@
 Expand /@ {$dm_{1,2\to 1} \sim B_1 \sim dS_1$ ≡ ($dS_1\ dS_2$) ~ $B_{1,2}$ ~ $dm_{2,1\to 1}$, $dS_1 \sim B_1 \sim d\Delta_{1\to 1,2}$ ≡ $d\Delta_{1\to 2,1} \sim B_{1,2}$ ~ ($dS_1\ dS_2$)}]

Out[∘]= {15.9375, {True, True}}

Quasi-triangular axiom 1:

In[∘]:= Timing@HL[$R_{1,2} \sim B_1 \sim d\Delta_{1->1,3}$ ≡ ($R_{1,4}\ R_{3,2}$) ~ $B_{2,4}$ ~ $dm_{2,4\to 2}$]

Out[∘]= {0.578125, True}

Quasi-triangular axiom 2:

In[∘]:= Timing@HL[(($d\Delta_{1->1,2}\ R_{3,4}$) ~ $B_{1,2,3,4}$ ~ ($dm_{1,3\to 1}\ dm_{2,4\to 2}$)) ≡ (($d\Delta_{1\to 2,1}\ R_{3,4}$) ~ $B_{1,2,3,4}$ ~ ($dm_{3,1\to 1}\ dm_{4,2\to 2}$))]

Out[∘]= {4.73438, True}

The Drinfel'd element inverse property, $(u_1\, \overline{u}_2) \sim B_{1,2} \sim \mathrm{dm}_{1,2\to 1} \equiv \mathbb{E}[0, 0, 1]$:

In[∘]:= Timing@
 HL[(($R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}$) ($R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j}$)) ~ $B_{i,j}$ ~ $dm_{i,j\to i}$ ≡ 𝔼[0, 0, 1]]

Out[∘]= {2.21875, True}

The ribbon element v satisfies $v^2 = S(u)\, u$. The spinner C=uv^{-1}. It is convenient to compute $z = S(u)\, u^{-1}$ which is something easy.

In[∘]:= Timing@Block[{$k = 2}, $(((R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}) \sim B_i \sim dS_i)\ (R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j})) \sim B_{i,j} \sim dm_{i,j\to i}$]

Out[∘]= $\left\{2.71875, \mathbb{E}\left[0, 0, \frac{1}{B_i} + \frac{a_i \in}{B_i} + \frac{a_i^2 \in^2}{2 B_i} + O[\in]^3\right]\right\}$

In[∘]:= Timing@Block[{$k = 2}, HL /@ { $(C_i \bar{C}_j) \sim B_{i,j} \sim dm_{i,j\to i} \equiv \mathbb{E}[0, 0, 1]$, $(\bar{C}_i \bar{C}_j) \sim B_{i,j} \sim dm_{i,j\to i} \equiv (((R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}) \sim B_i \sim dS_i)\ (R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j})) \sim B_{i,j} \sim dm_{i,j\to i}$}]

Out[∘]= {3.04688, {True, True}}

Reidemeister 2:

In[∘]:= Timing[HL[# ≡ 𝔼[0, 0, 1]] & /@ { $(\bar{R}_{1,2} R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to 1}\ dm_{2,4\to 2})$, $(R_{1,2} \bar{R}_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to 1}\ dm_{2,4\to 2})$ }]

Out[∘]= {3.46875, {True, True}}

Cyclic Reidemeister 2:

In[∘]:= Timing@HL[$(R_{1,4} \bar{R}_{5,2} \bar{C}_3) \sim B_{2,4} \sim dm_{2,4\to 2} \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{1,5} \sim dm_{1,5\to 1} \equiv \bar{C}_1$]

Out[∘]= {1.53125, True}

Reidemeister 3:

In[∘]:= Timing@HL[$((R_{1,2} R_{4,3} R_{5,6}) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3}) \equiv ((R_{1,6} R_{2,3} R_{4,5}) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3})$]

Out[∘]= {3.375, True}

Relations between the four kinks:

In[∘]:= Timing[HL /@ { $Kink_i \equiv (R_{3,1} C_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to i}$, $\overline{Kink}_j \equiv (\bar{R}_{3,1} \bar{C}_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to j}$, $(Kink_i\ \overline{Kink}_j) \sim B_{i,j} \sim dm_{i,j\to 1} \equiv \mathbb{E}[0, 0, 1]$ }]

Out[∘]= {3.48438, {True, True, True}}

The Trefoil

```
In[∘]:= Timing@Block[{$k = 1},
  Z = R1,5 R6,2 R3,7 C̄4 K̄ink8 K̄ink9 K̄ink10;
  Do[Z = Z~B1,r~dm1,r→1, {r, 2, 10}];
  {Simplify /@ Z, Simplify /@ (Z~B1~b2t1 /. T1 → T)}]
```

Out[∘]= $\left\{9.90625, \left\{\mathbb{E}\left[0, 0, \frac{B_1}{1 - B_1 + B_1^2} + \left(B_1 \left(-B_1 + 2 B_1^2 + 2 B_1^4 + a_1 \left(-1 + B_1 - B_1^3 + B_1^4\right) - 2 x_1 y_1 - B_1^3 \left(3 + 2 x_1 y_1\right)\right) \in\right) \Big/ \left(1 - B_1 + B_1^2\right)^3 + O[\in]^2\right], \mathbb{E}\left[0, 0, \frac{T}{1 - T + T^2} + \left(T \left(T \left(-1 + 2 T - 3 T^2 + 2 T^3\right) + 2 \left(-1 + T - T^3 + T^4\right) a_1 - 2 \left(1 + T^3\right) x_1 y_1\right) \in\right) \Big/ \left(1 - T + T^2\right)^3 + O[\in]^2\right]\right\}\right\}$

Program

In[∘]:=
```
Define[kR_{i,j} = R_{i,j} ~ B_{i,j} ~ (b2t_i b2t_j) /. t_{i|j} → t,
 kR̄_{i,j} = R̄_{i,j} ~ B_{i,j} ~ (b2t_i b2t_j) /. {t_{i|j} → t, T_{i|j} → T},
 km_{i,j→k} = (t2b_i t2b_j) ~ B_{i,j} ~ dm_{i,j→k} ~ B_k ~ b2t_k /. {t_k → t, T_k → T, τ_{i|j} → 0},
 kC_i = C_i ~ B_i ~ b2t_i /. T_i → T,
 kC̄_i = C̄_i ~ B_i ~ b2t_i /. T_i → T,
 kKink_i = Kink_i ~ B_i ~ b2t_i /. {t_i → t, T_i → T},
 kKink̄_i = Kink̄_i ~ B_i ~ b2t_i /. {t_i → t, T_i → T}]
```

In[∘]:=
```
Timing@Block[{$k = 1},
  Z = kR_{1,5} kR_{6,2} kR_{3,7} kC̄_4 kKink̄_8 kKink̄_9 kKink̄_10;
  Do[Z = Z ~ B_{1,r} ~ km_{1,r→1}, {r, 2, 10}];
  Simplify /@ Z]
```

Out[∘]=

$$\left\{4.70313,\ \mathbb{E}\left[0,\ 0,\ \frac{T}{1-T+T^2}+\left(T\left(T\left(-1+2T-3T^2+2T^3\right)+2\left(-1+T-T^3+T^4\right)a_1-2\left(1+T^3\right)x_1 y_1\right)\epsilon\right)\Big/\left(1-T+T^2\right)^3+O[\epsilon]^2\right]\right\}$$

RVK, rot, Z from 2016-09/OneSmidgen.nb. See also local version in this folder.

Program

Some details of the code below are at http://drorbn.net/bbs/show?shot=Dror-160920-151350.jpg.

Program

In[∘]:=
```
RVK::usage =
  "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
    xs and a length 2n list of rotation numbers rots. Crossing
    sites are indexed 1 through 2n, and rots⟦k⟧ is the rotation
    between site k-1 and site k. RVK is also a casting operator
    converting to the RVK presentation from other knot presentations.";
```

Program

In[∘]:=
```
RVK[pd_PD] := PP_RVK@Module[{n, xs, x, rots, front = {0}, k},
    n = Length@pd; rots = Table[0, {2 n}];
    xs = List @@ pd /. x_X :> If[PositiveQ@x, Xp[x⟦4⟧, x⟦1⟧], Xm[x⟦2⟧, x⟦1⟧]];
    For[k = 0, k < 2 n, ++k,
     If[k == 0 ∨ FreeQ[front, -k],
      front = Flatten[front /. k → Catch[xs /. {
            Xp[k + 1, l_] | Xm[l_, k + 1] :> Throw[{l, k + 1, 1 - l}],
            Xp[l_, k + 1] | Xm[k + 1, l_] :> (++rots⟦l⟧;
              Throw[{1 - l, k + 1, l}])
           }]]],
      If[MatchQ[front, {___, k, ___, -k, ___}], --rots⟦k + 1⟧]
     ]
    ];
    RVK[xs, rots]
   ];
RVK[K_] := RVK[PD[K]];
```

In[∘]:=
```
RVK[Knot[3, 1]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=
```
RVK[{Xm[4, 1], Xm[6, 3], Xm[2, 5]}, {0, 0, 0, -1, 0, 0}]
```

Program

In[∘]:=
```
rot[_, 0] = 𝔼[0, 0, 1];
rot[i_, n_] := Module[{j},
   rot[i, n] = If[n > 0, rot[i, n - 1] kC_j, rot[i, n + 1] \overline{kC}_j] ~B_{i,j}~km_{i,j→i}];
```

In[∘]:=
```
{rot[i, 3], rot[i, -3]}
```

Out[∘]=

$$\left\{\mathbb{E}\left[0,\,0,\,T^{3/2}-3\left(T^{3/2}\,a_i\right)\epsilon+\frac{9}{2}\,T^{3/2}\,a_i^2\,\epsilon^2+O[\epsilon]^3\right],\ \mathbb{E}\left[0,\,0,\,\frac{1}{T^{3/2}}+\frac{3\,a_i\,\epsilon}{T^{3/2}}+\frac{9\,a_i^2\,\epsilon^2}{2\,T^{3/2}}+O[\epsilon]^3\right]\right\}$$

Program

In[∘]:=
```
Z[K_] := Z[RVK@K];
Z[rvk_RVK] := (*Z[rvk] =*)
 PP_"Z"@Module[{todo, n, rots, ξ, done, st, cx, ξ1, i, j, k, k1, k2, k3},
   {todo, rots} = List @@ rvk;
   AppendTo[rots, 0];
   n = Length[todo];
   ξ = 𝔼[0, 0, 1];
   done = {0};
   st = Range[0, 2 n + 1];
   While[todo =!= {},
    {cx} = MaximalBy[todo, Length[done ⋂ {#[[1]], #[[2]], #[[1]] - 1, #[[2]] - 1}] &, 1];
    {i, j} = List @@ cx;
    ξ1 = Switch[Head[cx],
      Xp, (kR_{i,j} \overline{kKink}_k) ~B_{j,k}~km_{j,k→j},
      Xm, (\overline{kR}_{i,j} kKink_k) ~B_{j,k}~km_{j,k→j}
     ];
    ξ1 = (rot[k, rots[[i]]] ξ1) ~B_{k,i}~km_{k,i→i}; rots[[i]] = 0;
    ξ1 = (ξ1 rot[k, rots[[i + 1]]]) ~B_{i,k}~km_{i,k→i}; rots[[i + 1]] = 0;
    ξ1 = (rot[k, rots[[j]]] ξ1) ~B_{k,j}~km_{k,j→j}; rots[[j]] = 0;
    ξ1 = (ξ1 rot[k, rots[[j + 1]]]) ~B_{j,k}~km_{j,k→j}; rots[[j + 1]] = 0;
    ξ *= ξ1;
    If[MemberQ[done, i], ξ = ξ~B_{i,i+1}~km_{i,i+1→i}; st = st /. st[[i + 2]] → st[[i + 1]]];
    If[MemberQ[done, i - 1], ξ = ξ~B_{st[[i]],i}~km_{st[[i]],i→st[[i]]}; st = st /. st[[i + 1]] → st[[i]]];
    If[MemberQ[done, j], ξ = ξ~B_{j,j+1}~km_{j,j+1→j}; st = st /. st[[j + 2]] → st[[j + 1]]];
    If[MemberQ[done, j - 1], ξ = ξ~B_{st[[j]],j}~km_{st[[j]],j→st[[j]]}; st = st /. st[[j + 1]] → st[[j]]];
    done = done ⋃ {i - 1, i, j - 1, j};
    todo = DeleteCases[todo, cx]
   ];
   Simplify /@ (ξ /. {x_0 → x, y_0 → y, a_0 → a})
  ]
```

Knot

In[*]:= **$k = 1; Timing@Z@Knot[10, 100]**

Knot

Out[*]= $\{983.234, \mathbb{E}[0, 0, T^4/(1-4T+9T^2-12T^3+13T^4-12T^5+9T^6-4T^7+T^8) + (T^4(-6+2T^{16}+4a(-2+14T-51T^2+120T^3-203T^4+258T^5-246T^6+152T^7-152T^9+246T^{10}-258T^{11}+203T^{12}-120T^{13}+51T^{14}-14T^{15}+2T^{16})-8xy-440T^9(-1+xy)-4T^{15}(3+2xy)+8T^8(-97+21xy)+8T^7(131+21xy)-20T^6(57+22xy)+T^{14}(37+48xy)+T(44+48xy)-8T^{11}(2+61xy)+8T^5(127+68xy)-2T^{13}(35+78xy)+4T^{10}(-39+136xy)-T^2(167+156xy)+T^{12}(79+324xy)+T^3(410+324xy)-T^4(733+488xy))\in)/(1-4T+9T^2-12T^3+13T^4-12T^5+9T^6-4T^7+T^8)^3+O[\in]^2]\}$

In[*]:= **EndProfile[];**

Profile

In[*]:= **PrintProfile[]**

Profile

```
Out[*]= ProfileRoot is root. Profiled time: 1069.53
  (      1)    0.156/ 983.235 above Z
  (    163)    0.706/  73.748 above B
  (    147)        0/   0.047 above CF
  (      2)        0/       0 above RVK
  (     17)    0.016/   2.468 above Boot[1]
  (     18)    0.124/   7.641 above Boot[2]
  (      4)    0.015/   0.047 above Boot[3]
  (      1)        0/   2.344 above Boot[4]
Exp: called 247969 times, time in 625.943/696.494
  ( 247969)  625.943/ 696.494 under Together
  ( 245998)   38.153/  70.551 above CF
QZip: called 304 times, time in 209.497/871.153
  (    304)  209.497/ 871.153 under B
  (    912)    0.657/ 629.416 above CF
  (    304)   10.839/  32.240 above Zip
Together: called 247969 times, time in 72.842/769.336
  ( 247969)   72.842/ 769.336 under CF
  ( 247969)  625.943/ 696.494 above Exp
CF: called 247969 times, time in 53.551/822.887
  ( 245998)   38.153/  70.551 under Exp
  (    912)   14.741/ 122.873 under LZip
  (    147)        0/   0.047 under ProfileRoot
  (    912)    0.657/ 629.416 under QZip
  ( 247969)   72.842/ 769.336 above Together
Zip: called 2760 times, time in 49.408/149.258
  (    304)    6.157/  27.734 under LZip
  (    304)   10.839/  32.240 under QZip
  (   2152)   32.412/  89.284 under Zip
  (   2760)   10.566/  10.566 above Collect
  (   2152)   32.412/  89.284 above Zip
LZip: called 304 times, time in 46.162/196.769
  (    304)   46.162/ 196.769 under B
  (    912)   14.741/ 122.873 above CF
```

```
  (   304)    6.157/  27.734 above Zip
Collect: called 2760 times, time in 10.566/10.566
  (  2760)   10.566/  10.566 under Zip
B: called 304 times, time in 1.157/1069.08
  (    72)    0.310/ 982.986 under Z
  (   163)    0.706/  73.748 under ProfileRoot
  (    33)    0.031/   2.545 under Boot[1]
  (    29)    0.095/   7.487 under Boot[2]
  (     3)        0/   0.595 under Boot[3]
  (     4)    0.015/   1.718 under Boot[4]
  (   304)   46.162/ 196.769 above LZip
  (   304)  209.497/ 871.153 above QZip
Z: called 1 times, time in 0.156/983.235
  (     1)    0.156/ 983.235 under ProfileRoot
  (    72)    0.310/ 982.986 above B
  (     3)        0/   0.093 above Boot[1]
Boot[2]: called 23 times, time in 0.154/8.203
  (    18)    0.124/   7.641 under ProfileRoot
  (     5)    0.030/   0.562 under Boot[2]
  (    29)    0.095/   7.487 above B
  (     5)    0.030/   0.562 above Boot[2]
Boot[4]: called 6 times, time in 0.063/4.516
  (     1)        0/   2.344 under ProfileRoot
  (     3)        0/       0 under Boot[3]
  (     2)    0.063/   2.172 under Boot[4]
  (     4)    0.015/   1.718 above B
  (     1)        0/   0.563 above Boot[3]
  (     2)    0.063/   2.172 above Boot[4]
Boot[1]: called 27 times, time in 0.016/3.578
  (     3)        0/   0.093 under Z
  (    17)    0.016/   2.468 under ProfileRoot
  (     7)        0/   1.017 under Boot[1]
  (    33)    0.031/   2.545 above B
  (     3)        0/       0 above Boot[0]
  (     7)        0/   1.017 above Boot[1]
Boot[3]: called 5 times, time in 0.015/0.61
  (     4)    0.015/   0.047 under ProfileRoot
  (     1)        0/   0.563 under Boot[4]
  (     3)        0/   0.595 above B
  (     3)        0/       0 above Boot[4]
Boot[0]: called 3 times, time in 0./0.
  (     3)        0/       0 under Boot[1]
RVK: called 2 times, time in 0./0.
  (     2)        0/       0 under ProfileRoot
```